Designation: E 1926 – 98

Standard Practice for Computing International Roughness Index of Roads from Longitudinal Profile Measurements[1]

This standard is issued under the fixed designation E 1926; the number immediately following the designation indicates the year of original adoption or, in the case of revision, the year of last revision. A number in parentheses indicates the year of last reapproval. A superscript epsilon (ϵ) indicates an editorial change since the last revision or reapproval.

1. Scope

1.1 This practice covers the mathematical processing of longitudinal profile measurements to produce a road roughness statistic called the International Roughness Index (IRI).

1.2 The intent is to provide a standard practice for computing and reporting an estimate of road roughness for highway pavements.

1.3 This practice is based on an algorithm developed in The International Road Roughness Experiment sponsored by a number of institutions including the World Bank and reported in two World Bank Technical Papers **(1)(2)**.[2] Additional technical information is provided in two TRB papers **(3)(4)**.

1.4 The values stated in SI units are to be regarded as the standard. The inch-pound units given in parentheses are for information only.

1.5 *This standard does not purport to address all of the safety concerns, if any, associated with its use. It is the responsibility of the user of this standard to establish appropriate safety and health practices and determine the applicability of regulatory limitations prior to use.*

2. Referenced Documents

2.1 *ASTM Standards:*

E 177 Practice for Use of the Terms Precision and Bias in ASTM Test Methods[3]

E 867 Terminology on Vehicle—Pavement Systems[4]

E 950 Test Method for Measuring the Longitudinal Profile of Traveled Surfaces with an Accelerometer Established Inertial Profiling Reference[4]

E 1082 Test Method for Measurement of Vehicular Response to Traveled Surface Roughness[4]

E 1170 Practice for Simulating Vehicular Response to Longitudinal Profiles of a Vehicular Traveled Surface[4]

E 1215 Specification for Trailers Used for Measuring Vehicular Response to Road Roughness[4]

E 1364 Test Method for Measuring Road Roughness by Static Rod and Level Method[4]

E 1656 Guide for Classification of Automated Pavement Condition Survey Equipment[4]

3. Terminology

3.1 *Definitions*:

3.1.1 Terminology used in this practice conforms to the definitions included in Terminology E 867.

3.2 *Definitions of Terms Specific to This Standard:*

3.2.1 *international roughness index (IRI)*, *n*—an index computed from a longitudinal profile measurement using a quarter-car simulation (see Practice E 1170) at a simulation speed of 80 km/h (50 mph).

3.2.1.1 *Discussion*—IRI is reported in either meters per kilometer (m/km) or inches per mile (in./mile). (Note—1 m/kg = 63.36 in./mile.)

3.2.2 *longitudinal profile measurement*, *n*—a series of elevation values taken at a constant interval along a wheel track.

3.2.2.1 *Discussion*—Elevation measurements may be taken statically, as with rod and level (see Test Method E 1364 or inclinometer, or dynamically, as with an inertial profiler (see Test Method E 950).

3.2.3 *mean roughness index (MRI)*, *n*—the average of the International Roughness Index values for the right and left wheel tracks.

3.2.3.1 *Discussion*—Units are in metres per kilometer or inches per mile.

3.2.4 *traveled surface roughness*—the deviations of a surface from a true planar surface with characteristics dimensions that affect vehicle dynamics, ride quality, dynamic loads, and drainage, for example, longitudinal profile, transverse profile, and cross slope.

3.2.5 *true international roughness index*, *n*—the value of International Roughness Index that would be computed for a longitudinal profile measurement with the constant interval approaching zero.

3.2.6 *wave number*, *n*—the inverse of wavelength.

3.2.6.1 *Discussion*—Wave number, sometimes called spatial frequency, typically has units of cycle/m or cycle/ft.

---

[1] This practice is under the jurisdiction of ASTM Committee E-17 on Vehicle-Pavement Systems and is the direct responsibility of Subcommittee E17.33 on Methodology for Analyzing Pavement Roughness.

Current edition approved Jan. 10, 1998. Published March 1999.

[2] The boldface numbers given in parentheses refer to a list of references at the end of the text.

[3] *Annual Book of ASTM Standards*, Vol 14.02.

[4] *Annual Book of ASTM Standards*, Vol 04.03.

3.2.7 *wheel track*, *n*—a line or path followed by the tire of a road vehicle on a traveled surface.

4. Summary of Practice

4.1 The practice presented here was developed specifically for estimating road roughness from longitudinal profile measurements.

4.2 Longitudinal profile measurements for one wheel track are transformed mathematically by a computer program and accumulated to obtain the IRI. The profile must be represented as a series of elevation values taken at constant intervals along the wheel track.

4.3 The IRI scale starts at zero for a road with no roughness and covers positive numbers that increase in proportion to roughness. Fig. 1 associated typical IRI values with verbal descriptors from World Bank Technical Paper Number 46 **(2)** for roads with bituminous pavement, and Fig. 2 shows similar associations for roads with earth or gravel surfaces.

5. Significance and Use

5.1 This practice provides a means for obtaining a quantative estimate of a pavement property defined as roughness using longitudinal profile measuring equipment.



FIG. 1 Road Roughness Estimation Scale for Paved Roads With Asphaltic Concrete or Surface Treatment (Chipseal)



Recently bladed surface of fine gravel, or soil surface with excellent longitudinal and transverse profile (usually found only in short length)

Ride comfortable up to 80-100 km/h, aware of gentle undulation or swaying. Negligible depressions (e.g. < 5mm/3m) and no potholes.

Ride comfortable up to 70-80 km/h, aware of sharp movements and some wheel bounce. Frequent shallow-mode rate depressions or shallow potholes (e.g. 6-30mm/3m with frequency 5-10 per 50m). Moderate corrugations (e.g. 6-20mm/0.7-1.7m).

Ride comfortable at 50 km/h (or 40-70 km/h on specific sections). Frequent moderate transverse depressions (e.g. 20-40mm/3-5m) or occasional deep depressions or potholes (e.g. 40-80mm/3m with frequency less than 5 per 50m). Strong corrugations (e.g. 20mm/0.7-1.5m).

Ride comfortable at 30-40km/h. Frequent deep transverse depressions and/or potholes (e.g. 40-80mm at freq. 5-10 per 50m); or occasional very deep depressions (e.g. 80mm/1-5m with frequency less than 5 per 50m) with other shallow depressions. Not possible to avoid all the depressions except the worst.

Ride comfortable at 20-30km/h. Speeds higher than 40-50 km/h would cause extreme discomfort, and possible damage to the car. On a good general profile; frequent deep depressions and/or potholes (e.g. 40-80mm/1-5 at frequency 10-15 per 50m) and occasional very deep depressions (e.g. > 80mm/0.6-2m). On a poor general profile; frequent moderate defects and depressions (e.g. poor earth surface).

FIG. 2 Road Roughness Estimation Scale for Unpaved Roads with Gravel or Earth Surfaces

5.1.1 The IRI is portable in that it can be obtained from longitudinal profiles obtained with a variety of instruments.

5.1.2 The IRI is stable with time because true IRI is based on the concept of a true longitudinal profile, rather than the physical properties of a particular type of instrument.

5.2 Roughness information is a useful input to the pavement management systems (PMS) maintained by transportation agencies.

5.2.1 The IRI for the right wheel track is the measurement of road surface roughness specified by the Federal Highway Administration (FHWA) as the input to their Highway Performance Monitoring System (HPMS).

5.2.2 When profiles are measured simultaneously for both traveled wheel tracks, then the MRI (average of the IRI for each profile) is considered to be a better measure of road surface roughness than the IRI for either wheel track.

NOTE 1—The MRI scale is identical to the IRI scale.

5.3 IRI can be interpreted as the output of an idealized response-type measuring system (see Test Method E 1082 and

Specification E 1215), where the physical vehicle and instrumentation are replaced with a mathematical model. The units of slope correspond to accumulated suspension motions (for example, metres), divided by the distance traveled (for example, kilometres).

5.4 IRI is a useful calibration reference for response-type systems that estimate roughness by measuring vehicular response (see Test Method E 1082 and Specification E 1215).

5.5 IRI can also be interpreted as average absolute slope of the profile, filtered mathematically to modify the amplitudes associated with different wavelengths **(3)**.

6. Longitudinal Profile Measurement

6.1 The longitudinal profile measurements can be obtained from equipment that operate in a range of speeds from static to highway traffic speeds.

6.2 The elevation profile measuring equipment used to collect the longitudinal profile data used in this practice must have sufficient accuracy to measure the longitudinal profile attributes that are essential to the computation of the International Roughness Index.

6.2.1 A repeating sine wave of the following wavelengths and peak-to-peak amplitudes in the absence of any other profile roughness will produce the following IRI values:

Amplitude, mm	Wavelength, m	IRI, m/km
25.4	91.4	0.0222

Amplitude, in.	Wavelength, ft	IRI, in./mi
1.00	300	1.42

7. Computation of International Roughness Index (IRI)

7.1 This practice consists of the computation of IRI from an algorithm developed in the International Road Roughness Experiment and described in the World Bank Technical Papers 45 and 46 **(1)(2)**. Additional technical information provided in two TRB papers **(3)(4)**.

7.2 A FORTRAN version of this algorithm has been implemented as described in Ref **(3)**.

7.2.1 This practice presents a sample computer program" IRISMP" for the computation of the IRI from the recorded longitudinal profile measurement.

7.2.1.1 The computer program IRISMP is a general computer program which accepts the elevation profile data set as input and then calculates the IRI values for that profile data set.

7.2.1.2 A listing of the IRISMP computer program for the computation of IRI is included in this practice as Appendix X1.

7.2.1.3 A provision has been made in the computer program listing (Appendix X1) for the computation of IRI from recorded longitudinal profile measurements in either SI or inch-pound units.

7.2.2 The input to the sample IRI computer program is an ASCII profile data set stored in a 1X,F8.3,1X,F8.3 Fortran format. In this format, the profile data appears as a multi-row, two column array with the left wheel path profile data points in Column 1 and the right wheel path points in Column 2. The profile data point interval is discretionary. However the quality of the IRI values computed by this algorithm is a function of the data point interval.

7.2.2.1 If the input to the IRI computer program is in SI units, the elevation profile data points are scaled in millimetres with the least significant digit being equal to 0.001 mm.

7.2.2.2 If the input to the IRI computer program is in inch-pound units, the elevation profile data points are scaled in inches with the least significant digit being equal to 0.001 in.

7.3 The distance interval over which the IRI is computed is discretionary, but shall be reported along with the IRI results.

7.4 Validation of the IRI program is required when it is installed. Provision for the IRI program installation validation has been provided in this practice.

7.4.1 The sample profile data set TRIPULSE.ASC has been provided in SI units in Appendix X2 for validation of the computer program installation.

7.4.2 Using the sample profile data set TRIPULSE.ASC (Appendix X2) as input to the IRI computer program (Appendix X1), an IRI value of 4.42 mm/m was computed as shown in Appendix X3 for a profile data point interval of .15 m (.5 ft) and a distance interval equal to 15 m of the profile data set.

8. Report

8.1 Include the following information in the report for this practice:

8.1.1 *Profile Measuring Device*—The Class of the profile measuring device used to make the profile measurement as defined in Test Method E 950 and Test Method E 1364 shall be included in the report.

8.1.2 *Longitudinal Profile Measurements*—Report data from the profile measuring process shall include the date and time of day of the measurement, the location of the measurement, the lane measured, the direction of the measurement, length of measurement, and the descriptions of the beginning and ending points of the measurement. The recorded wheel track (left, right, or both) must also be included.

8.1.3 *IRI Resolution*—The number of digits after the decimal point depends on the choice of units. If the units are m/km, then results should be reported with two digits after the decimal point. If the units are in./mi, then the IRI results should be reported to a resolution of 0.1 in./mi.

9. Precision and Bias

9.1 The precision and bias of the computed IRI is limited by the procedures used in making the longitudinal profile measurement. Guidelines for measuring longitudinal profile are provided in Test Method E 950 and Test Method E 1364.

9.2 For the effects of the precision and bias of the measured profile on the computed IRI, see precision and bias in Appendix X1.

10. Keywords

10.1 highway performance monitoring system; HPMS; international roughness index; IRI; longitudinal profile; pavement management systems; pavement roughness; PMS

APPENDIXES

(Nonmandatory Information)

X1. PRECISION AND BIAS

X1.1 *Precision*:

X1.1.1 The precision of the computed IRI is limited by the procedures used in making the longitudinal profile measurement. Guidelines for measuring longitudinal profile are provided in Test Method E 950 and Test Method E 1364.

X1.1.2 IRI precision depends on the interval between adjacent profile elevation measures (see Test Method E 950 and Test Method E 1364). Reducing the interval typically improves the precision. An interval of 0.3 m (12 in.) or smaller is recommended. For some surface types, a shorter interval will improve precision. More information about the sensitivity of IRI to sample interval is provided in Ref **(3)**.

X1.1.3 IRI precision is roughly equivalent to the precision of the slope obtained from the longitudinal profile measurements, for distances ranging from approximately 1.5 m (5 ft) to about 25 m (80 ft). For example, a relative error on profile elevation of 1.0 mm over a distance of 10 m corresponds to a slope error of 0.1 mm/m, or 0.1 m/km (6.3 in/mi).

X1.1.4 IRI precision is limited by the degree to which a wheel track on the road can be profiled. Errors in locating the wheel track longitudinally and laterally can influence the IRI values, because the IRI will be computed for the profile of the wheel track as measured, rather than the wheel track as intended. These effects are reduced by using longer profiles.

X1.1.5 Computational errors due to round-off are typically about two orders of magnitude smaller than those due to limitations in the profile measuring process, and can be safely ignored.

X1.2 *Bias*:

X1.2.1 The bias of the computed IRI is typically limited by the procedures used in making the longitudinal profile measurement. Guidelines for measuring longitudinal profile are provided in Test Method E 950 and Test Method E 1364.

X1.2.2 IRI bias depends on the interval between adjacent profile elevation measures. An interval of 0.3 m (12 in.) or smaller is recommended. Shorter intervals improve precision but have little effect on bias. More information about the sensitivity of IRI to sample interval is provided in Ref **(3)**.

X1.2.3 Many forms of measurement error cause an upward bias in IRI. (The reason is that variations in profile elevation due to measurement error are usually uncorrelated with the profile changes.) Some common sources of positive IRI bias are: height-sensor round-off, mechanical vibrations in the instrument that are not corrected and electronic noise. Bias is reduced by using profiler instruments that minimize these errors.

X1.2.4 Inertial profiler systems (see Test Method E 950) include one or more filters that attenuate long wavelengths (low wave numbers). If the cut-off wavelength is too short, then the IRI computed from the profile will have a negative bias. A cut off wavelength of 91.4 m/cycle (300 ft/cycle) is considered sufficiently long.

NOTE X1.1—Profiles obtained with static methods are generally not filtered, and therefore this source of bias is not relevant for them.

X1.2.5 The measures from some inertial profilers are processed during measurement to attenuate short wavelengths and prevent aliasing. The effect is to smooth the profile measurement. If a smoothing filter is used and it affects wavelengths longer than 1 m (3.3 ft), then the computed IRI will have a negative bias.

NOTE X1.2—If the profiler includes a smoothing filter that affects wavelengths shorter than 1 m (3.3 ft) and longer than 250 mm (10 in.), no more smoothing is required during the computation of IRI.

X2. INTERNATIONAL ROUGHNESS INDEX COMPUTER PROGRAM

X2.1 Included in this appendix is the coding in Fortran language for a computer subroutine, SUBROUTINE IRI, Fig. X2.1, which calculates the International Roughness Index as prescribed by this practice. A sample main program is also included which when executed, prompts the user for the name of a data file containing the profile data to be processed and the parameters needed by the subroutine to compute the IRI. The subroutine is called and returns the computed IRI values to the main program which then displays them.

X2.2 The sample program can process data files containing two profile tracks in either SI or inch-pound units. For SI data, the program assumes the input amplitudes are stored in millimetre units; if inch-pound, inches. For the sample program, the maximum length road section that can be processed is limited to 1058 sample pairs.

X2.3 The sample data file shown in Fig. X2.12 and Fig. X2.13 is in SI units (mm) and contains one hundred and one profile data point pairs. The tracks are identical. The sampling interval for the data is 0.15 m.

```
C================================================
Listing 0. Fortran code for sample 'main' program.
c================================================

c          l.lnk:
c          irismp+     'MAIN' program plus IRI subroutines
c            ;
c
c   Sample IRI Fortran Computer Program
c==========================================
c- Sample program to read a data file containing two tracks of
c- road profile elevation data into a "DATA" array, call SUBR IRI
c- and print a final report of International Roughness Index.
c- If the input profile data is in English units, the elevation values
c- are converted from inch to mm units and the sampling interval, from
c- feet to meters; the computed IRI values are returned as m/km and
c- converted to in/mi.
c- (SUBROUTINE IRI is called to perform the IRI computation as
c- prescribed by this practice.)
c==========================================
c
      PROGRAM IRISMP
c
c-----------------------------------
c
c     DELT ---------------> DX
c     DATA(1, 1058) ------> PROF(NPTS), left track
c     DATA(2, 1058) ------> PROF(NPTS), right track
c     AVIRIL -------------> IRI, left track
c     AVIRIR -------------> IRI, right track
c     AVEIRI -------------> == (AVIRIL+AVIRIR)/2.
c     UNITSC -------------> see UNITSC (SUBR IRI)
c
c==========================================

      REAL          DATA(2, 1058), DELT
      REAL          DINTVL, SECLEN
      REAL          BASE, UNITSC, PROF(1058),SCLFAC
      REAL          AVIRIL, AVIRIR, AVEIRI
c
      BYTE          UNITS, ANSWER
      CHARACTER   KNAME*12
c
      INTEGER       NPTS, NREC, I, J
```

FIG. X2.1 Sample Fortran Program Using Subroutine IRI to Compute International Ride Index

```
c-----------------------------------
c
   NREC = 0
c
   WRITE (*, 1000)
1000  FORMAT (/'1Enter Data File Name (in single quotes)'/
  1          '    ("TRIPULSE.ASC" in example):   '$)
c
   READ(*,*)KNAME(1:12)
c
   WRITE(*,1010)
1010  FORMAT(/'1Is road profile data in SI or inch-pound units?'/
  1       '      Type "S" or "I", (cr)  :    '$)
c
   READ(*,1020)UNITS
1020  FORMAT(A1)
c
   IF(UNITS.EQ.'I'.OR.UNITS.EQ.'i') WRITE(*,1030)
1030  FORMAT(/' Enter length of section, feet '/
  1                  '     (82 ft in example)      :    '$)
c
   IF(UNITS.EQ.'S'.OR.UNITS.EQ.'s') WRITE(*,1040)
1040  FORMAT(/' Enter length of section, meters '/
  1                  '     (25 m in example)      :    '$)
c
   READ(*,1050)SECLEN
1050  FORMAT(F10.0)
c
   IF(UNITS.EQ.'I'.OR.UNITS.EQ.'i') WRITE(*,1060)
1060  FORMAT(/' Enter sampling interval, ft. '/
  1                  '     (.82 ft in example)      :    '$)
c
   IF(UNITS.EQ.'S'.OR.UNITS.EQ.'s') WRITE(*,1070)
1070  FORMAT(/' Enter sampling interval, meters '/
  1                  '     (.25 m in example)      :    '$)
c
   READ(*,1050)DELT
c
   IF(UNITS.EQ.'I'.OR.UNITS.EQ.'i')SECLEN=SECLEN/3.281  !ft --> m
c
   IF(UNITS.EQ.'I'.OR.UNITS.EQ.'i')DELT=DELT/3.281  !ft --> m
c
   NPTS=SECLEN/DELT              !number of elevation pairs in DATA array.
*
```

FIG. X2.1 Sample Fortran Program Using Subroutine IRI to Compute International Ride Index *(continued)*

```
c
     WRITE(*,1075)
1075  FORMAT(/' Is Input Profile Pre-smoothed (Y or N) ? '$)
     READ(*,1020)ANSWER
     BASE = .250                  ! meters
     IF(ANSWER .NE. 'N'.AND. ANSWER .NE. 'n') BASE = 0.
c-----------------------------
c- Open input file and read profile elevations
c  into 'DATA' array:
c
     OPEN(UNIT=2,FILE=KNAME(1:12),FORM='FORMATTED')
c
c-----------------------------
c
     UNITSC = 1.
c
     DO 20 I = 1,NPTS
c
     READ(2,1080)(DATA(J, I), J = 1, 2)    ! read 'NPTS' data pairs
c                                  into DATA array
1080  FORMAT(2(1X, F8.3))
c
20    CONTINUE
c-----------------------------------
c- Call subroutine to calculate International Roughness Index:
c
c  Copy left wheelpath profile to 'PROF' array:
c
     DO 30 I = 1, NPTS
     PROF(I) = DATA(1, I)
     IF(UNITS .EQ. 'I' .OR. UNITS .EQ. 'i')
    & PROF(I)=PROF(I)*25.4          ! in --> mm
c
30 CONTINUE
c---------------------------------------------------
     CALL IRI (PROF, NPTS, DELT, BASE, UNITSC, AVIRIL)
c-----------------------------------------------------
c
c  Copy right wheelpath profile to PROF array:
c
c
     NPTS=SECLEN/DELT              !number of elevation pairs in DATA array.
c
     DO 40 I = 1, NPTS
     PROF(I) = DATA(2, I)
     IF(UNITS .EQ. 'I' .OR. UNITS .EQ. 'i')
    & PROF(I)=PROF(I)*25.4          ! in --> mm
c
40    CONTINUE
```

FIG. X2.1 Sample Fortran Program Using Subroutine IRI to Compute International Ride Index *(continued)*

```
c
c-------------------------------------------------------
      CALL IRI (PROF, NPTS, DELT, BASE, UNITSC, AVIRIR)
c--------------------------------------------------------
c
      AVEIRI = (AVIRIL + AVIRIR)/2.
c
      DINTVL = NPTS*DELT                      ! length of profile section
c
c-----------------------------------
c- Output computed International Roughness Index:
c
      IF(UNITS.EQ.'S'.OR.UNITS.EQ.'s')GOTO 200
c
      SCLFAC=5280./25.4/3.281         ! = 63.36
      WRITE(*, 2010)
     1 AVIRIL*SCLFAC, AVIRIR*SCLFAC ,AVEIRI*SCLFAC, DINTVL*3.281
c
      WRITE(*, 2010)
2010  FORMAT(//////
     1          6X' IRI, left                = ',F10.1,' in/mi'//
     2          6X' IRI, right               = ',F10.1,' in/mi'////
     3          6X' International Roughness Index = ',F10.1,' in/mi'//
     4       6X' Distance    =  ',F6.1,' feet'/)
c
      GOTO 210
c---------------------------------------
c
200   CONTINUE
      WRITE(*, 2020)
     1 AVIRIL, AVIRIR, AVEIRI, DINTVL

2020  FORMAT(//////
     1          6X' IRI, left                = ',F10.2' mm/m'//
     2          6X' IRI, right               = ',F10.2,' mm/m'////
     3          6X' International Roughness Index = ',F10.2,' mm/m'//
     4          6X' Distance    =  ',F6.1,' meters'/)
c
c-----------------------------------
210   CONTINUE
c
      END
```

FIG. X2.1 Sample Fortran Program Using Subroutine IRI to Compute International Ride Index *(continued)*

```
C==========================================================
      SUBROUTINE IRI(PROF, NSAMP, DX, BASE, UNITSC, AVEIRI)
C==========================================================
C  Filter a longitudinal road profile and calculate IRI.
C
C  <-> PROF   REAL      On input, an array of profile height values.
C                       On output, an array of filtered profile values.
C  <-> NSAMP  INTEGER   Number of data values in array PROF. Filtered
C                       profile always has fewer points than original.
C  --> DX     REAL      Distance step between profile points (m).
C  --> BASE   REAL      Distance covered by moving average (m).
C                       Use .250 for unfiltered profile input, and 0.0
C                       for pre-smoothed profiles (e.g. K. J. Law data).
C  --> UNITSC REAL      Product of two scale factors: (1) metres per unit
C                       of profile height, and (2) IRI units of slope.
C                       Ex: height is inches, slope will be in/mi.
C                       UNITSC = (.0254 m/in)*(63360 in/mi) = 1609.34
c
C  <-- AVEIRI REAL      The average IRI for the entire profile.
c
      INTEGER           I, I11, IBASE, J, NSAMP
      REAL              AMAT, AVEIRI, BASE, BMAT, CMAT, DX
      REAL              UNITSC, V, XIN, PROF, SFPI, ST, PR
c      DIMENSION        AMAT(4, 4), BMAT(4), CMAT(4), PR(4),
c    &            ST(4,4), XIN(4), PROF(NSAMP+2)
      DIMENSION         AMAT(4, 4), BMAT(4), CMAT(4), PR(4),
     &                  ST(4,4), XIN(4), PROF(1058)
c
C  Set parameters and arrays.
      CALL SETABC(653.0, 63.3, 6.0, 0.15, AMAT, BMAT, CMAT)
      CALL SETSTM(DX/(80./3.6), AMAT, BMAT, ST, PR)
      IBASE = MAX(INT(BASE/DX + 0.5), 1)
      SFPI = UNITSC/(DX*IBASE)

C  Initialize simulation variables based on profile start.
      I11 = MIN(INT(11./DX + .5) +1, NSAMP)
      XIN(1) = UNITSC*(PROF(I11) - PROF(1))/(DX*I11)
      XIN(2) = 0.0
      XIN(3) = XIN(1)
      XIN(4) = 0.0

C  Convert to averaged slope profile, with IRI units.
      NSAMP = NSAMP - IBASE
      DO 10 I = 1, NSAMP
   10 PROF(I) = SFPI*(PROF(I + IBASE) - PROF(I))
```

FIG. X2.1 Sample Fortran Program Using Subroutine IRI to Compute International Ride Index *(continued)*

```
C  Filter profile.
        CALL STFILT(PROF, NSAMP, ST, PR, CMAT, XIN)
c
C  Compute IRI from filtered profile.
        AVEIRI = 0.0
        DO 20 I = 1, NSAMP
  20   AVEIRI = AVEIRI + ABS(PROF(I))
        AVEIRI = AVEIRI/NSAMP
c
        RETURN
        END

c                    Listing 2. Code to set model matrices.
C=====================================================================
        SUBROUTINE SETABC(K1, K2, C, MU, AMAT, BMAT, CMAT)
C=====================================================================
C  Set the A, B and C matrices for the 1/4 car model.
C
C  -->    K1   REAL          Kt/Ms = normalized tire spring rate, (1/s/s)
C  -->    K2   REAL          Ks/Ms = normalized suspension spring rate (1/s/s)
C  -->    C    REAL          Ks/Ms = normalized suspension damper rate (1/s)
C  -->    MU   REAL          Ks/Ms = normalized unsprung mass (-)
C  <--    AMAT REAL          The 4x4 A matrix.
C  <--    BMAT REAL          The 4x1 B matrix.
C  <--    CMAT REAL          The 4x1 C matrix.

        INTEGER        I, J
        REAL           AMAT, BMAT, CMAT, K1, K2, C, MU
        DIMENSION      AMAT(4, 4), BMAT(4), CMAT(4)

C  Set default for all matrix elements to zero.
        DO 10  J = 1, 4
          BMAT(J) = 0
          CMAT(J) = 0
          DO 10  I = 1, 4
10          AMAT(I, J) = 0

C  Put 1/4 car model parameters into the A Matrix.
        AMAT(1, 2) = 1.
        AMAT(3, 4) = 1.
        AMAT(2, 1) = -K2
        AMAT(2, 2) = -C
        AMAT(2, 3) = K2
        AMAT(2, 4) = C
        AMAT(4, 1) = K2/MU
        AMAT(4, 2) = C/MU
        AMAT(4, 3) = -(K1 + K2)/MU
        AMAT(4, 4) = -C/MU
```

FIG. X2.1 Sample Fortran Program Using Subroutine IRI to Compute International Ride Index *(continued)*

```
C  Set the B matrix for road input through tire spring.
         BMAT(4) = K1/MU

C  Set the C matrix to use suspension motion as output.
         CMAT(1) = -1
         CMAT(3) = 1

         RETURN
         END

c                   Listing 3. Code to set state transition matrix.
C==========================================================
     SUBROUTINE SETSTM(DT, A, B, ST, PR)
C==========================================================
C
C  Compute ST and PR arrays.  This requires INVERT for matrix inversion.
C
C  --> DT        REAL    Time step (sec)
C  --> A                 REAL           The 4x4 A matrix.
C  --> B                 REAL           The 4x4 B matrix.
C  <-- ST        REAL            4x4 state transition matrix.
C  <-- PR        REAL            4x1 partial response vector.

         INTEGER         I, ITER, J, K
         LOGICAL         MORE
         REAL            A, A1, A2, B, DT, PR, ST, TEMP
         DIMENSION       A(4, 4), A1(4, 4), A2(4, 4), B(4)
         DIMENSION       PR(4),ST(4, 4), TEMP(4, 4)

         DO 20 J = 1,4
          DO 10 I = 1,4
           A1(I, J) = 0
10         ST(I, J) = 0
          A1(J, J) = 1.
20        ST(J, J) = 1.

C  Calculate the state transition matrix ST = exp(dt*A) with a Taylor
C  series. A1 is the previous term, A2 is the next one.
         ITER = 0
  30 ITER = ITER + 1
         MORE = .FALSE.
         DO 40 J = 1,4
          DO 40 I = 1,4
           A2(I, J) = 0
           DO 40 K = 1,4
  40      A2(I, J) = A2(I, J) + A1(I, K) * A(K, J)
```

FIG. X2.1 Sample Fortran Program Using Subroutine IRI to Compute International Ride Index *(continued)*

```
      DO 50 J = 1,4
       DO 50 I = 1,4
        A1(I, J) = A2(I, J)*DT/ITER
        IF (ST(I, J) + A1(I, J) .NE. ST(I, J)) MORE = .TRUE.
 50     ST(I, J) = ST(I, J) + A1(I, J)
   IF (MORE) GO TO 30

C  Calculate particular response matrix: PR = A**-1*(ST-I)*B
      CALL INVERT(A, 4)
      DO 60 I = 1,4
       PR(I) = 0.0
       DO 60 K = 1,4
 60     PR(I) = PR(I) - A(I, K)*B(K)
      DO 90 J = 1,4
       DO 70 I = 1, 4
        TEMP(J, I) = 0.0
        DO 70 K = 1,4
 70      TEMP(J, I) = TEMP(J, I) + A(J, K)*ST(K, I)
       DO 80 K = 1, 4
 80    PR(J) = PR(J) + TEMP(J, K)*B(K)
 90    CONTINUE
      RETURN
      END
C
C                 Listing 4. Code to filter profile.
C============================================================
      SUBROUTINE STFILT(PROF, NSAMP, ST, PR, C, XIN)
C============================================================
C  Filter profile using matrices ST, PR, and
C
C <->  PROF   REAL          Input profile. Replaced by the output.
C -->  NSAMP  INTEGER       Number of data values in array PROF.
C -->  ST     REAL          4x4 state transition matrix.
C -->  PR     REAL          4x1 partial response vector.
C -->  C      REAL          4x1 output definition vector.
C -->  XIN    real          Initial values of filter variables.

      INTEGER I, J, K, NSAMP,ii
      REAL          C, PR, PROF, ST, X, XIN, XN
c      DIMENSION    C(4), PR(4), PROF(NSAMP+2), ST(4, 4)
      DIMENSION    C(4), PR(4), PROF(1058), ST(4, 4)
      DIMENSION    X(4), XIN(4), XN(4)
c     DIMENSION    C(4), PR(4), PROF(1000), ST(4, 4) X(4), XIN(4), XN(4)

C  Initialize simulation variables.
   DO 10 I = 1,4
 10     X(I) = XIN(I)
```

FIG. X2.1 Sample Fortran Program Using Subroutine IRI to Compute International Ride Index *(continued)*

```
C  Filter profile using the state transition algorithm.
   DO 40 I = 1, NSAMP
       DO 20 J = 1, 4
        XN(J) = PR(J)*PROF(I)
        DO 20 K = 1, 4
 20      XN(J) = XN(J) + X(K)*ST(J, K)

          X(1)=XN(1)
          X(2)=XN(2)
          X(3)=XN(3)
          X(4)=XN(4)

       PROF(I) = X(1)*C(1) + X(2)*C(2) + X(3)*C(3) + X(4)*C(4)

 40    CONTINUE

   RETURN
   END


C                    Listing 5. Code to invert matrix.
C===================================================================
   SUBROUTINE INVERT(Y1, N)
C===================================================================
C  This routine will store the inverse of NxN matrix Y1 in matrix YINV.
C  It was copied from "Numerical Recipes."
C
C  Y1   --> Real    The matrix to be inverted.
C  YINV --> Real    The inverse of matrix Y1.

   INTEGER     N, INDX, I, J
   REAL*4      Y1, YINV, D, A
   DIMENSION   Y1(N, N), YINV(4, 4), INDX(4), A(4, 4)

   DO 8 I = 1, N
    DO 9 J = 1, N
 9    A(I, J) = Y1(I, J)
 8   CONTINUE
   DO 10 I = 1, N
    DO 20 J = 1, N
 20   YINV(I, J) = 0.0
    YINV(I, I) = 1.0
 10  CONTINUE
   CALL LUDCMP(A, INDX, D)
   DO 30 J = 1, N
 30  CALL LUBKSB(A, INDX, YINV(1, J))
   DO 40 I = 1, N
    DO 50 J = 1, N
 50   Y1(I ,J) = YINV(I, J)
```

FIG. X2.1 Sample Fortran Program Using Subroutine IRI to Compute International Ride Index *(continued)*

```
 40  CONTINUE
     RETURN
     END

C==================================================================
     SUBROUTINE LUDCMP(A, INDX, D)
C==================================================================
C  This routine was copied from "Numerical Recipes" for matrix
C  inversion.
C
     INTEGER       N, INDX, NMAX, I, J, IMAX, K
     REAL*4      A, TINY, VV, D, AAMAX, SUM, DUM
     PARAMETER    (NMAX = 100, TINY = 1.0E-20, N = 4)
     DIMENSION    A(N, N), INDX(N), VV(NMAX)

     D = 1.0
     DO 10 I = 1, N
      AAMAX = 0.0
      DO 20 J = 1, N
 20     IF(ABS(A(I,J)).GT.AAMAX) AAMAX=ABS(A(I,J))
      IF(AAMAX.EQ.0.0) PAUSE 'Singular matrix'
      VV(I) = 1.0/AAMAX
 10  CONTINUE
     DO 30 J = 1, N
      DO 40 I = 1, J-1
           SUM = A(I, J)
       DO 50 K = 1, I-1
 50      SUM = SUM - A(I, K)*A(K, J)
       A(I, J) = SUM
 40    CONTINUE
      AAMAX = 0.0
      DO 60 I = J, N
       SUM = A(I, J)
       DO 70 K = 1, J-1
 70      SUM = SUM - A(I, K)*A(K, J)
       A(I, J) = SUM
       DUM = VV(I)*ABS(SUM)
       IF(DUM.GE.AAMAX)THEN
        IMAX = I
        AAMAX = DUM
       ENDIF
 60    CONTINUE
      IF(J.NE.IMAX)THEN
       DO 80 K = 1, N
        DUM = A(IMAX, K)
        A(IMAX, K) = A(J, K)
        A(J, K) = DUM
 80      CONTINUE
       D = -D
```

FIG. X2.1 Sample Fortran Program Using Subroutine IRI to Compute International Ride Index *(continued)*

```
     VV(IMAX) = VV(J)
    ENDIF
    INDX(J) = IMAX
    IF(A(J, J).EQ.0.0) A(J, J) = TINY
    IF(J.NE.N)THEN
     DUM = 1.0/A(J, J)
     DO 90 I = J+1, N
90      A(I, J) = A(I, J)*DUM
    ENDIF
30  CONTINUE
   RETURN
   END

C======================================================================
   SUBROUTINE LUBKSB(A, INDX, B)
C======================================================================
C  This routine was copied from "Numerical Recipes" for matrix
C  inversion.

   INTEGER      N, INDX, I, II, LL, J
   REAL*4      A, B, SUM
   PARAMETER    (N = 4)
   DIMENSION    A(N, N), INDX(N), B(N)

   II = 0
   DO 10 I = 1, N
    LL = INDX(I)
    SUM = B(LL)
    B(LL) = B(I)
    IF(II.NE.0)THEN
     DO 20 J = II, I-1
20      SUM = SUM - A(I, J)*B(J)
    ELSEIF(SUM.NE.0)THEN
     II = I
    ENDIF
    B(I) = SUM
10  CONTINUE
   DO 30 I = N, 1, -1
    SUM = B(I)
    IF(I.LT.N)THEN
     DO 40 J = I+1, N
40      SUM = SUM - A(I, J)*B(J)
    ENDIF
    B(I) = SUM/A(I, I)
30  CONTINUE
   RETURN
   END
```

FIG. X2.1 Sample Fortran Program Using Subroutine IRI to Compute International Ride Index *(continued)*

0.000	0.000
0.000	0.000
0.000	0.000
0.000	0.000
0.000	0.000
2.500	2.500
5.000	5.000
7.500	7.500
10.000	10.000
12.500	12.500
15.000	15.000
17.500	17.500
20.000	20.000
17.500	17.500
15.000	15.000
12.500	12.500
10.000	10.000
7.500	7.500
5.000	5.000
2.500	2.500
0.000	0.000
0.000	0.000
0.000	0.000

... (pad with zeros to make a total of 101 numerical data)

0.000	0.000
0.000	0.000
0.000	0.000
0.000	0.000
0.000	0.000
0.000	0.000

NOTE 1—Elevations are metric units (mm). The profile consists of identical right and tracks, each consisting of zero elevations everywhere except the triangular 'pulse' from 1.0 to 5.0 m peaking at 20.0 mm. The interval between elevations is 0.15 m. This data set may be used as a test of the user's implementation of IRI standard computation.

FIG. X2.12 Sample Load Profile Input Data Set, TRIPULSE.ASC

```
Enter Data File Name (in single quotes)
('TRIPULSE.ASC' in example)      :  'TRIPULSE.ASC'

Is road profile data in SI or inch-pound units?

Type 'S' or 'I'              :  S

Enter length of section, m (ft)
(25 meters in example)        :  15

Enter sample interval, meters    :  .15

Is Input Profile presmoothed?   :  Y



IRI, left wheel path        =   4.42 mm/m

IRI, right wheel path        =   4.42 mm/m

International Roughness Index =   4.42 mm/m


Distance                =   14.8 meters
```

FIG. X2.13 Input/Output for RNSMP sample program using data input file 'TRIPULSE.ASC'

REFERENCES

(**1**) Sayers, M.W., Gillespie, T.D., Queiroz, C.A.V., "The International Road Roughness Experiment", *World Bank Technical Paper Number 45*, 1986.

(**2**) Sayers, M.W., Gillespie, T.D., Paterson, W.D.O., "Guidelines for Conducting and Calibrating Road Roughness Measurements", *World Bank Technical Paper Number 46*, 1986.

(**3**) Sayers, M.W.," On the Calculation of IRI from Longitudinal Profile," *Transportation Research Record 1501*, Transportation Research Board, pp. 1–12.

(**4**) Sayers, M.W., "Two Quarter-Car Models for Defining Road Roughness: IRI and HRI., *Transportation Research Record 1215*, 1989, pp. 165–172.